


Chapter 9: Efficient portfolio theorems

Th. Warin



Notation (1)

N risky assets.

Vector of expected returns:

$$E(r) = \begin{bmatrix} E(r_1) \\ E(r_2) \\ \vdots \\ E(r_N) \end{bmatrix}$$

Variance-covariance matrix:

$$S = \begin{bmatrix} \sigma_{11} & \sigma_{12} & \cdots & \sigma_{1N} \\ \sigma_{21} & \sigma_{22} & \cdots & \sigma_{2N} \\ \vdots & & & \\ \sigma_{N1} & \sigma_{N2} & \cdots & \sigma_{NN} \end{bmatrix}$$

Notation (2)

A *portfolio* of risky assets is a set of proportions x_i which sum to 1.

$$x = \begin{bmatrix} x_1 \\ x_2 \\ \vdots \\ x_N \end{bmatrix}, \quad \sum_{i=1}^N x_i = 1$$

The portfolio expected return is:


$$E(r_x) = x^T \cdot R \equiv \sum_{i=1}^N x_i E(r_i)$$

Notation (3)

The portfolio variance is

$$\sigma_x^2 = x^T S x = \sum_{i=1}^N \sum_{j=1}^N x_i x_j \sigma_{ij}$$

Feasible Portfolios




FIVE PROPOSITIONS ABOUT THE ENVELOPE PORTFOLIOS

Five propositions about envelope portfolios

Proposition 1 (Merton 1973): All envelope portfolios solve the following equation:

$$x = \frac{S^{-1} \{E(r) - c\}}{\sum S^{-1} \{E(r) - c\}}$$

Here c is an arbitrary constant.



Proposition 1 gives a method of finding the tangency portfolio.

Computing two efficient portfolios

	A	B	C	D	E	F	G	H	
1	CALCULATING THE EFFICIENT FRONTIER								
2	Variance-covariance matrix					Expected returns E(r)	Expected minus constant E(r) - c		
3	0.40	0.03	0.02	0.00		0.06	0.02	<-- =F3-\$B\$8	
4	0.03	0.20	0.00	-0.06		0.05	0.01	<-- =F4-\$B\$8	
5	0.02	0.00	0.30	0.03		0.07	0.03	<-- =F5-\$B\$8	
6	0.00	-0.06	0.03	0.10		0.08	0.04	<-- =F6-\$B\$8	
7									
8	Constant	0.04							
9									
10	Computing an envelope portfolio with constant = 0								
11	z					Envelope portfolio x			
12	0.1019	{=MMULT(MINVERSE(A3:D6),F3:F6)}				0.0540	<-- =A12/SUM(\$A\$12:\$A\$15)		
13	0.5657					0.2998			
14	0.1141					0.0605			
15	1.1052					0.5857			
16					Sum	1.0000	<-- =SUM(F12:F15)		
17									
18	Computing an envelope portfolio with constant = 0.04								
19	z					Envelope portfolio y			
20	0.0330	<-- {=MMULT(MINVERSE(A3:D6),G3:G6)}				0.0423	<-- =A20/SUM(\$A\$20:\$A\$23)		
21	0.1959					0.2514			
22	0.0468					0.0601			
23	0.5035					0.6462			
24					Sum	1.0000	<-- =SUM(F20:F23)		

Here we compute the efficient portfolio in 2 steps.

In step 1 we compute a portfolio z:

$$z = S^{-1} \{E(r) - c\}$$

In step 2, we normalize z to sum to 1, computing two efficient portfolios, x and y.

Proposition 2 (Merton): The convex combination of any two envelope portfolios is also an envelope portfolio.

Thus: if x and y are envelope portfolios, so is

$$\lambda x + (1 - \lambda) y = \left\{ \begin{array}{c} \lambda x_1 + (1 - \lambda) y_1 \\ \dots \\ \lambda x_N + (1 - \lambda) y_N \end{array} \right\}$$

	A	B	C	D	E	F	G
1	SOME c's CAN LEAD TO INEFFICIENT PORTFOLIOS The portfolio x determined by the constant c=0.11 is inefficient						
2	Variance-covariance matrix					Expected returns E(r)	
3	0.40	0.03	0.02	0.00		0.06	
4	0.03	0.20	0.00	-0.06		0.05	
5	0.02	0.00	0.30	0.03		0.07	
6	0.00	-0.06	0.03	0.10		0.08	
7							
8	Constant	0.11					
9							
10	Computing an envelope portfolio with constant = 0.11						
11	z					Envelope portfolio x	
12	-0.0876	<-- {=MMULT(MINVERSE(A3:D6),F3:F6-B8)}				0.0755	<-- {=A12:A15/SUM(A12:A15)}
13	-0.4514					0.3893	
14	-0.0710					0.0613	
15	-0.5495					0.4739	
16				Sum		1.0000	<-- =SUM(F12:F15)
17							
18							
19	E(r _x)	0.0662					
20	σ _x	0.1944					
21							
22							
23							
24							
25							
26							
27							
28							
29							
30							
31							
32							

Proposition 1 defines a procedure for finding an envelope portfolio.

This portfolio can be efficient or inefficient.

Computing the envelope (1)

- Propositions 1 and 2 show how to compute the envelope (the efficient frontier)
- First—compute two envelope portfolios x and y . Do this with two arbitrary constants c_1 and c_2 .
- Second—compute the mean and standard deviation of all convex combinations of x and y .


Computing the envelope (2)

	A	B	C	D	E	F	G	H	I	J	K
26	E(x)	0.0693			E(y)	0.0710	<-- {=MMULT(TRANSPOSE(F20:F23),F3:F6)}				
27	Var(x)	0.0367			Var(y)	0.0398	<-- {=MMULT(MMULT(TRANSPOSE(F20:F23),A3:D6),F20:F23)}				
28	Sigma(x)	0.1917			Sigma(y)	0.1995	<-- =SQRT(F27)				
29											
30		Cov(x,y)	0.0376	<-- {=MMULT(MMULT(TRANSPOSE(F12:F15),A3:D6),F20:F23)}							
31		Corr(x,y)	0.9842	<-- =C30/(B28*F28)							
32											
33											
34	A single portfolio calculation										
35	Proportion of x	0.3									
36	$E(r_p)$	7.05%	<-- =B35*B26+(1-B35)*F26								
37	σ_p	19.65%	<-- =SQRT(B35^2*B27+(1-B35)^2*F27+2*B35*(1-B35)*C30)								

- ❑ Portfolios x and y are envelope portfolios.
- ❑ Rows 26:31 compute the relevant statistics
- ❑ Rows 35:37 compute the expected return and standard deviation of a single portfolio in which proportion of $x = 0.3$ and the proportion of $y = 0.7$.

In the next slide we repeat the computations in a **Data Table** to compute the whole envelope.

Computing the envelope (3)



Any two constants lead to same efficient frontier

- An efficient frontier is a combination of two efficient portfolios.
- Doesn't matter which c 's you use to compute the two efficient portfolios—you get the same efficient frontier .

	A	B	C	D	E	F	G	H
1	FINDING EFFICIENT PORTFOLIOS IN ONE STEP							
2	Variance-covariance matrix					Expected returns E(r)		
3	0.40	0.03	0.02	0.00		0.06		
4	0.03	0.20	0.00	-0.06		0.05		
5	0.02	0.00	0.30	0.03		0.07		
6	0.00	-0.06	0.03	0.10		0.08		
7								
8	Constant	0.05						
9								
10	Envelope portfolio							
11	0.0314	<-- {=MMULT(MINVERSE(A3:D6),F3:F6-B8)/SUM(MMULT(MINVERSE(A3:D6),F3:F6-B8))}						
12	0.2059							
13	0.0597							
14	0.7031							
15								
16	Portfolio expected return, $E(r_x)$	7.26%	<-- =SUMPRODUCT(A11:A14,F3:F6)					
17	Portfolio standard deviation, σ_x	21.21%	<-- {=SQRT(MMULT(MMULT(TRANSPOSE(A11:A14),A3:D6),A11:A14))}					

Creative use of array functions: Use to find efficient portfolios in one step (cells A11:A14).

Proposition 3 (Black, 1972): If portfolio y is envelope, then all other portfolios are related to y through the following linear relationship:

$$E(r_x) = c + \beta_x [E(r_y) - c]$$

where

$$\beta_x = \frac{\text{Cov}(x, y)}{\sigma_y^2}$$

Proposition 4 (Black): If the linear relationship below holds, then portfolio y is an envelope portfolio.

$$E(r_x) = c + \beta_x [E(r_y) - c]$$

where

$$\beta_x = \frac{\text{Cov}(x, y)}{\sigma_y^2}$$


Proposition 5: If there exists a risk-free asset with return r_f , then the standard security market line (SML) relationship holds:

$$E(r_x) = r_f + \beta_x [E(r_x) - r_f]$$

where

$$\beta_x = \frac{\text{cov}(x, M)}{\sigma_M^2}$$

	A	B	C	D	E	F	G
1	IF $c = r_f$ AND THE OPTIMIZING PORTFOLIO IS EFFICIENT, THEN THE ENVELOPE PORTFOLIO IS OPTIMAL						
	The portfolio x determined by the constant $c= 4\%$ is optimal						
2	Variance-covariance matrix					Expected returns	
3	0.40	0.03	0.02	0.00		0.06	
4	0.03	0.20	0.00	-0.06		0.05	
5	0.02	0.00	0.30	0.03		0.07	
6	0.00	-0.06	0.03	0.10		0.08	
7							
8	Constant	0.04					
9							
10	Computing an envelope portfolio with constant = 0.04						
11	z					Envelope portfolio x	
12	0.0330	<-- {=MMULT(MINVERSE(A3:D6),F3:F6-B8)}				0.0423	<-- {=A12:A15/SUM(A12:A15)}
13	0.1959					0.2514	
14	0.0468					0.0601	
15	0.5035					0.6462	
16					Sum	1.0000	<-- =SUM(F12:F15)
17							
18							
19	$E(r_x)$	0.0710					
20	σ_x	0.1995					
21							
22							
23							
24							
25							
26							
27							
28							
29							
30							



	A	B	C	D	E	F	G
1	ILLUSTRATING PROPOSITIONS 3-5						
2	Dates	Asset 1	Asset 2	Asset 3	Asset 4		Efficient portfolio w
3	1	-6.63%	-2.49%	-4.27%	11.72%		-1.82%
4	2	8.53%	2.44%	-3.15%	-8.33%		0.99%
5	3	1.79%	4.46%	1.92%	19.18%		5.47%
6	4	7.25%	17.90%	-6.53%	-7.41%		3.65%
7	5	0.75%	-8.22%	-1.76%	-1.44%		-2.51%
8	6	-1.57%	0.83%	12.88%	-5.92%		2.16%
9	7	-2.10%	5.14%	13.41%	-0.46%		4.14%
10							
11	Mean	1.15%	2.87%	1.79%	1.05%	<-- =AVERAGE(E3:E9)	1.73%
12							
13	Variance-covariance matrix						
14		Asset 1	Asset 2	Asset 3	Asset 4		
15	Asset 1	0.0024	0.0019	-0.0015	-0.0024		
16	Asset 2	0.0019	0.0056	-0.0007	-0.0016		
17	Asset 3	-0.0015	-0.0007	0.0057	-0.0005		
18	Asset 4	-0.0024	-0.0016	-0.0005	0.0094		
19							
20	Finding an efficient portfolio w						
21	Constant	0.50%					
22							
23	Asset 1	0.3129					
24	Asset 2	0.2464					
25	Asset 3	0.2690					
26	Asset 4	0.1717					

Cells B15:E18 contain the formula
 {=MMULT(TRANSPOSE(B3:E9-B11:E11),B3:E9-B11:E11)/7}

Cells B23:B26 contain the formula
 {=MMULT(MINVERSE(B15:E18),TRANSPOSE(B11:E11)-B21)/SUM(MMULT(MINVERSE(B15:E18),TRANSPOSE(B11:E11)-B21))}

The efficient portfolio in B23:B26 depends on the constant in cell B21. The returns in G3:G9 are the annual returns of this efficient portfolio.

For example, cell G3:

$$-1.82\% = 0.3129 \cdot -6.63\% + 0.2464 \cdot -2.49\% + 0.2690 \cdot -4.27\% + 0.1717 \cdot 11.72\%$$

	A	B	C	D	E	F	G
29	Implementing propositions 3-5--finding the SML						
30	Step 1: Regress each asset's returns on those of the efficient portfolio w						
31		Asset 1	Asset 2	Asset 3	Asset 4		
32	Alpha	0.0024	-0.0047	-0.0002	0.0028	<-- =INTERCEPT(E3:E9,\$G\$3:\$G\$9)	
33	Beta	0.5284	1.9301	1.0490	0.4478	<-- =SLOPE(E3:E9,\$G\$3:\$G\$9)	
34	R-squared	0.0897	0.5241	0.1505	0.0167	<-- =RSQ(E3:E9,\$G\$3:\$G\$9)	
35							
36	Step 2: Regress the asset mean returns on their betas						
37	Intercept	0.005	<-- =INTERCEPT(B11:E11,B33:E33)				
38	Slope	0.0123	<-- =SLOPE(B11:E11,B33:E33)				
39	R-squared	1.0000	<-- =RSQ(B11:E11,B33:E33)				
40							
	Check Propositions 3 & 4: Step 2 coefficients should be:						
41	Intercept = c, Slope = $E(r_w) - c$						
42	Intercept = c ?	yes	<-- =IF(B36=B20,"yes","no")				
43	Slope = $E(r_w) - c$?	yes	<-- =IF(B38=G11-B21,"yes","no")				

Rows 30-34: Regressing each asset's returns on the efficient portfolio.

Rows 36-39: Regressing asset mean returns on their betas. **The regression is perfect.**

This is what Propositions 3-5 guarantee—because the regression is done with respect to an efficient portfolio.

CONCLUSION

Some philosophy (I)

- CAPM is the search for a linear return relation between expected return and portfolio risk .
- This is the meaning of the classic SML:
- Propositions 3,4 say: There is a linear relation if and only if portfolios are regressed on an efficient portfolio.

Philosophy (2)

- Classic SML tests are two-step regressions:
 - Pick an index M which might represent the market portfolio.
 - Step 1: Regress each asset's returns on the index M to find the asset's β :
$$r_{it} = \alpha_i + \beta_i r_{Mt}$$
 - Step 2: Regress the asset mean returns on the betas to “find” the SML

The actual market portfolio is the value-weighted portfolio of all risky assets.

$$\bar{r}_i = \gamma_0 + \gamma_1 \beta_i$$

- Question: What's the R^2 of the SML?

Philosophy (3): Roll (1977)

- From Propositions 3,4 we learn that an $R^2 = 100\%$ exists only if the index M is efficient.
- Therefore the real question: Is the true market portfolio efficient?
- Other questions of the same type:
 - Does God exist?
 - ???